```
> restart;
  with(PDEtools);
```

[*CanonicalCoordinates*, *ChangeSymmetry*, *CharacteristicQ*, *CharacteristicQInvariants*, *ConservedCurrentTest*, *ConservedCurrents*, *ConsistencyTest*, *D_Dx*, *DeterminingPDE*, *Eta_k*, *Euler*, *FromJet*, *InfinitesimalGenerator*, *Infinitesimals*, *IntegratingFactorTest*, *IntegratingFactors*, *InvariantEquation*, *InvariantSolutions*, *InvariantTransformation*, *Invariants*, *Laplace*, *Library*, *PDEplot*, *PolynomialSolutions*, *ReducedForm*, *SimilaritySolutions*, *SimilarityTransformation*, *Solve*, *SymmetrySolutions*, *SymmetryTest*, *SymmetryTransformation*, *TWSolutions*, *ToJet*, *build*, *casesplit*, *charstrip*, *dchange*, *dcoeffs*, *declare*, *diff_table*, *difforder*, *dpolyform*, *dsubs*, *mapde*, *separability*, *splitstrip*, *splitsys*, *undeclare*] **(1)**

Finite difference stencils for higher order derivatives

The purpose of this worksheet is show how one can deveolp stencils for higher order derivative appearing in partial differential equations. Our motivation is that most of the PDEs we are interested in involve second or higher order derivatives of the unknown function. For example, the prototypical parabolic PDE is

```
> parabolic := diff(u(t,x),t) - d*diff(u(t,x),x,x);
```

$$parabolic := \frac{\partial}{\partial t}\, u(t,x) - d\left(\frac{\partial^2}{\partial x^2}\, u(t,x)\right) \tag{2}$$

In this formula, we would like to replace both the spatial and temporal derivatives by some sort of finite difference. For the spatial derivative, we could try a stencil as follows

```
> N := 3;
  n := floor(N/2);
  stencil := diff(u(t,x),x,x) = add(beta[i]*u(t,x+i*h),i=-n..n);
```

$$N := 3$$

$$n := 1$$

$$stencil := \frac{\partial^2}{\partial x^2}\, u(t,x) = \beta_{-1}\, u(t, x-h) + \beta_0\, u(t,x) + \beta_1\, u(t, x+h) \tag{3}$$

As usual, we need to choose the beta[i] coefficients in order to make the Taylor series expansions of the LHS and RHS match to some order. In this example, there are three undetermine coefficient, so we can match Taylor series up to order 2. Here is the Taylor series expansion of the LHS - RHS of the stencil

```
> series1 := convert(series((lhs-rhs)(stencil),h=0,N),D);
```

$$series1 := \mathrm{D}_{2,2}(u)(t,x) - \beta_{-1}\, u(t,x) - \beta_1\, u(t,x) - \beta_0\, u(t,x) + \left(-\beta_1\, \mathrm{D}_2(u)(t,x) + \beta_{-1}\, \mathrm{D}_2(u)(t,x)\right) h + \left(-\frac{1}{2}\,\beta_1\, \mathrm{D}_{2,2}(u)(t,x) - \frac{1}{2}\,\beta_{-1}\, \mathrm{D}_{2,2}(u)(t,x)\right) h^2 + \mathrm{O}\left(h^3\right) \tag{4}$$

We re-organize the above into a polynomial in u and its derivatives (called series2):

```
> vars := [u(t,x),seq(D[2$i](u)(t,x),i=1..N-1)];
  series2 := collect(convert(series1,polynom),vars,'distributed');
```

$$vars := [u(t,x), D_2(u)(t,x), D_{2,2}(u)(t,x)]$$

$$series2 := (-\beta_{-1} - \beta_1 - \beta_0)\, u(t,x) + (-\beta_1 + \beta_{-1})\, h\, D_2(u)(t,x) + \left(1 + \left(-\frac{1}{2}\beta_1 - \frac{1}{2}\beta_{-1}\right) h^2\right) D_{2,2}(u)(t,x) \quad \textbf{(5)}$$

The coefficients in this must vanish, giving us a linear system for the beta's

```
> eqs := [coeffs(series2,vars)];
  beta_sol := solve(eqs);
  centered := factor(subs(beta_sol,stencil));
```

$$eqs := \left[-\beta_{-1} - \beta_1 - \beta_0, (-\beta_1 + \beta_{-1})\, h, 1 + \left(-\frac{1}{2}\beta_1 - \frac{1}{2}\beta_{-1}\right) h^2\right]$$

$$beta\_sol := \left\{h = h, \beta_{-1} = \frac{1}{h^2}, \beta_0 = -\frac{2}{h^2}, \beta_1 = \frac{1}{h^2}\right\}$$

$$centered := \frac{\partial^2}{\partial x^2} u(t,x) = -\frac{-u(t,x-h) + 2\,u(t,x) - u(t,x+h)}{h^2} \quad \textbf{(6)}$$

So this is our stencil. To determine the magnitude of the error, we can expand the RHS in a series about h = 0. We obtain

```
> series(rhs(centered),h);
```

$$D_{2,2}(u)(t,x) + \frac{1}{12} D_{2,2,2,2}(u)(t,x)\, h^2 + O(h^4) \quad \textbf{(7)}$$

The first term is what we want, and we see that the magnitude of the next term (the error) is O(h^2). So our stencil is accurate to O(h). The above stencil is called "centered" because it approximates the derivative by using an equal number of points on either side. The following procedure calculates an N point centered stencil to the r^th derivative of u. Here, N must be an odd integer greater that r, otherwise we get nonsense (why?). By default, it calcuates the spatial stencil, but by including the optional argument direction = temporal, it calculates a temporal stencil. The output is an equation with the stencil on the LHS and the leading order error on the RHS.

```
> centered_stencil := proc(r,N,{direction := spatial})
      local n, stencil, vars, beta_sol:
      n := floor(N/2):
      if (direction = spatial) then:
         stencil := D[2$r](u)(t,x) - add(beta[i]*u(t,x+i*h),i=-n..
  n);
         vars := [u(t,x),seq(D[2$i](u)(t,x),i=1..N-1)];
      else:
         stencil := D[1$r](u)(t,x) - add(beta[i]*u(t+i*h,x),i=-n..
  n);
         vars := [u(t,x),seq(D[1$i](u)(t,x),i=1..N-1)];
      fi:
      beta_sol := solve([coeffs(collect(convert(series(stencil,h,
  N),polynom),vars,'distributed'),vars)]):
      stencil := subs(beta_sol,stencil);
      if (direction = spatial) then:
         convert(stencil = convert(series(stencil,h,N+2),polynom),
  diff);
      else:
         subs(h=s,convert(stencil = convert(series(stencil,h,N+2),
  polynom),diff));
```

```
        fi:
  end proc:
```

Here are some examples of how the procedure works:

```
> centered_stencil(2,3);
  centered_stencil(1,3,direction=temporal);
  centered_stencil(5,9);
```

$$\frac{\partial^2}{\partial x^2} u(t,x) - \frac{u(t,x-h)}{h^2} + \frac{2\,u(t,x)}{h^2} - \frac{u(t,x+h)}{h^2} = -\frac{1}{12}\left(\frac{\partial^4}{\partial x^4} u(t,x)\right) h^2$$

$$\frac{\partial}{\partial t} u(t,x) + \frac{1}{2}\,\frac{u(t-s,x)}{s} - \frac{1}{2}\,\frac{u(t+s,x)}{s} = -\frac{1}{6}\left(\frac{\partial^3}{\partial t^3} u(t,x)\right) s^2$$

$$\frac{\partial^5}{\partial x^5} u(t,x) - \frac{1}{6}\,\frac{u(t,x-4\,h)}{h^5} + \frac{3}{2}\,\frac{u(t,x-3\,h)}{h^5} - \frac{13}{3}\,\frac{u(t,x-2\,h)}{h^5} + \frac{29}{6}\,\frac{u(t,x-h)}{h^5} - \frac{29}{6}\,\frac{u(t,x+h)}{h^5} + \frac{13}{3}\,\frac{u(t,x+2\,h)}{h^5} - \frac{3}{2}\,\frac{u(t,x+3\,h)}{h^5} + \frac{1}{6}\,\frac{u(t,x+4\,h)}{h^5} = \frac{13}{144}\left(\frac{\partial^9}{\partial x^9} u(t,x)\right) h^4 \tag{8}$$

An alternative to the centered stencil is the one-sided stencil where one approximates the derivative by using points to one side. It is easy to modify the above procedure to generate such a stencil (in this, N is an integer greater than r):

```
> onesided_stencil := proc(r,N,{direction := spatial})
      local stencil, vars, beta_sol:
      if (direction = spatial) then:
          stencil := D[2$r](u)(t,x) - add(beta[i]*u(t,x+i*h),i=0..
  N-1);
          vars := [u(t,x),seq(D[2$i](u)(t,x),i=1..N-1)];
      else:
          stencil := D[1$r](u)(t,x) - add(beta[i]*u(t+i*h,x),i=0..
  N-1);
          vars := [u(t,x),seq(D[1$i](u)(t,x),i=1..N-1)];
      fi:
      beta_sol := solve([coeffs(collect(convert(series(stencil,h,
  N),polynom),vars,'distributed'),vars)]):
      stencil := subs(beta_sol,stencil);
      if (direction = spatial) then:
          convert(stencil = convert(series(`leadterm`(stencil),h,
  N+1),polynom),diff);
      else:
          subs(h=s,convert(stencil = convert(series(`leadterm`
  (stencil),h,N+1),polynom),diff));
      fi:
  end proc:
```

Here are some examples:

```
> onesided_stencil(1,2,direction=temporal);
  onesided_stencil(2,3,direction=spatial);
  onesided_stencil(3,7,direction=temporal);
```

$$\frac{\partial}{\partial t} u(t,x) + \frac{u(t,x)}{s} - \frac{u(t+s,x)}{s} = -\frac{1}{2}\left(\frac{\partial^2}{\partial t^2} u(t,x)\right) s$$

$$\frac{\partial^2}{\partial x^2} u(t,x) - \frac{u(t,x)}{h^2} + \frac{2\,u(t,x+h)}{h^2} - \frac{u(t,x+2\,h)}{h^2} = -\left(\frac{\partial^3}{\partial x^3} u(t,x)\right) h$$

$$\frac{\partial^3}{\partial t^3} u(t,x) + \frac{49}{8}\,\frac{u(t,x)}{s^3} - \frac{29\,u(t+s,x)}{s^3} + \frac{461}{8}\,\frac{u(t+2\,s,x)}{s^3} - \frac{62\,u(t+3\,s,x)}{s^3} + \frac{307}{8}\,\frac{u(t+4\,s,x)}{s^3} - \frac{13\,u(t+5\,s,x)}{s^3} + \frac{15}{8}\,\frac{u(t+6\,s,x)}{s^3} = \frac{29}{15}\left(\frac{\partial^7}{\partial t^7} u(t,x)\right) s^4 \tag{9}$$

Beyond centered and one-sided stencils, we can generally define stencil on irregular or asymmetric lattices. The general rule is that you need at least r+1 points in your stencil to get an approximation to the r^th derivative.